Node Gardens

Introduction

A few months back, I was perusing one of my favorite web sites, www.levitated.net and ran across its creator, Jared Tarbell’s latest entry, The Node Garden (link). This just blew me away. In it he describes a system of interconnected “nodes” that, while independent, can communicate to nearby nodes in a sort of neural network. This idea was simple and elegant and really struck a chord in me. The graphics, as usual on levitated, are stunning. The pages linked to above include a set of functions used to create and animate the nodes. But, in my usual fashion, I was more inspired by the concepts presented, and went about coding my own version.

Jared and I are in touch from time to time and have shared and extended each other’s ideas in the past, so I knew he’d have no problem in me taking a crack at my own version. The results of this simple system are some of the coolest, most exciting files I’ve ever created, in my own opinion. The Node Garden itself is a simple framework, but has endless possibilities. I’ve spent so many hours changing node behavior, tweaking values, redefining connections, etc. A few people, in turn, have been inspired by some of my node gardens and have created their own, which then left me in awe and wanting to try some new concepts. It’s my hope that by sharing some of the concepts I used, as well as turning people onto Jared’s original pages, some even more incredible files will evolve.

Making Nodes

Note: Everything following is, for the most part, my own ideas. All of the code is stuff I dreamed up on my own. I suggest you check Jared’s for the levitated take on this, as he is the originator of the concept.

So, what is a node? It is simply a unit, an entity that can contain certain behaviors. Nodes can connect, interact, and communicate with other nodes. A node by itself generally does not do too much that is very interesting. It is the interaction of the system as a whole that draws you in.

We start out by making a node symbol in the library. It can be whatever shape, color or form you want. I used a small, white to black gradient circle, which looked pretty cool on a black background.

My node is 10 pixels in diameter. The gradient goes from pure white in the middle, at 100 percent alpha, to black with 0 alpha on the outside.

How to make a node:

1. Make your shape.

2. Select it.

3. Hit F8 to turn it into a movie clip. Select “Export for ActionScript” in the dialog. This will make it available to dynamically attach to the stage later. Name the symbol and the linkage name, “node”.

4. Delete the original instance that’s left on stage.

 Now we have a node available for attaching. Our first sample file will simply create a bunch of nodes and randomly place them on stage. This code would be placed in Frame one of the main time line:

numNodes = 50;

for(i=0;i<numNodes;i++){


node = attachMovie(“node”, “n”+i, i);


node._x = Math.random()*550;


node._y = Math.random()*400;


node._xscale=node._yscale=Math.random()*100+50;

}

There we have it, our first node garden. Actually, this looks like nothing more than a star field right now. That’s because it is pretty static. But each one of these “stars” is a movie clip, with all the capabilities that offers. Each one can be moved, scaled, faded, given functions to run on each frame to animate itself, respond to keyboard or mouse input, etc. We will explore a few possibilities here.

Interaction
Next up, we will start programming some node interaction. In order to do this, we need to have some form of communication between nodes. Down at the lowest level, communication will be one on one. One node communicating to another node. And, any node can potentially communicate to any other node, but doesn’t necessarily have to.

So, here is what we need to do:

1. Get two nodes together.

2. Have them decide whether or not they will communicate.

3. Have them communicate.

4. Get two more nodes and start over at 1. above.

The first challenge is to make sure we have every node paired up with every other node. To start with, we’ll put them all in a nice convenient place to refer to them – an array. Then we don’t have to worry about dynamically recreating their names each time, we just have each node in an array element, and loop through the array to hit them all. Our code so far becomes this:

numNodes = 50;

nodez = [];

for(i=0;i<numNodes;i++){


node = attachMovie(“node”, “n”+i, i);


nodez[i] = node;


node._x = Math.random()*550;


node._y = Math.random()*400;


node._xscale=node._yscale=Math.random()*100+50;

}

The line:

nodez = [];

is a shorthand way of creating an empty array. Same as saying:

nodez = new Array();

Then, the line:

nodez[i] = node;

adds a reference to the latest attached node into the next element of the array. Now we have nodez, which is an array holding all of our nodes, right at our fingertips.

Next, we need to twin up every node to every other node, one by one. The most obvious way to do this would be to take the first node, nodez[0], and loop through the array, matching it with each other one. Then take nodez[1], and do the same with that. This would give you a nested loop, like below:

for(i=0;i<nodez.length;i++){


nodeA = nodez[i];


for(j=0;j<nodez.length;j++){



nodeB = nodez[j];



// now match up nodeA and nodeB



// and have them communicate


}

}

That’s the obvious way, but has several flaws. The first one comes in the first time through the loops. At that point, both i and j will be equal to zero. So you will be matching nodez[0] to nodez[0], which isn’t really necessary. This will happen for every node, whenever i and j are equal.

The second bug comes on the second time through the first loop. The first time through, you successfully match nodez[0] with every other node on the list (including itself). Now i=1, so we are dealing with nodez[1]. We start the inner loop with j=0. So we are now matching nodez[1] with nodez[0]. But we already did that. Actually, each time we go through the loop, we are doing more and more unnecessary work, making matches that have already been made. In fact, the last time through, we are dealing with the last node, comparing it to all the other nodes, when all the other nodes have already compared themselves to it! This structure ends up doing more than twice the work needed!

OK, here’s the solution. Just a couple of minor changes:

for(i=0;i<nodez.length-1;i++){


nodeA = nodez[i];


for(j=i+1;j<nodez.length;j++){



nodeB = nodez[j];



// now match up nodeA and nodeB



// and have them communicate


}

}

First off, we change the first line so that i only goes up to nodez.length-1. This kills off the final loop, which we just saw was unnecessary. Then, we have our j loop start with a value of i+1. So, when i is 0, j will start with 1. When i is 22, j will start with 23. This not only keeps a node from matching with itself, but cuts out checking all the prior nodes which have already been matched.

All right, now we are ready for the second step which is:

2. Have them decide whether or not they will communicate.

Now, what the heck does that mean? How are they supposed to “decide”? Well, all a decision is is a conditional. If (something) then we will communicate. You decide what the condition is. It can be anything. That is one of the beautiful things about a node garden, is the variety you can program into it. If this node is red and that one is green, then communicate. If both nodes are above a certain size, communicate. If they are both moving in the same direction, or have similar charge (which is just a property you give them) then they should communicate.

Here’s an easy one. If they are close to each other, they should communicate. That’s about the most basic of them. You get two nodes, measure the distance between them, and if it’s less than a certain amount, they communicate. Here’s how we do that:

minDist = 80;

for(i=0;i<nodez.length-1;i++){


nodeA = nodez[i];


for(j=i+1;j<nodez.length;j++){



nodeB = nodez[j];



dx = nodeB._x – nodeA._x;



dy = nodeB._y – nodeA._y;



dist = Math.sqrt(dx*dx+dy*dy);



if(dist<minDist){




// communicate



}


}

}

Here, we set a value, minDist to 80. We find the distance between the two nodes on the x and y axis, use a bit of math to determine the actual distance, and then compare that to the minimum. If it is less, they communicate.

What is Communication???

I know you’ve been dying to ask this one. I keep saying that the nodes will communicate. What does that mean? Well, similar to the decision point, it can mean whatever you want it to. Communication generally means that some information is passed from one person or terminal to another. Here, we can generalize it a bit more and just say that there is some reaction that occurs between the two, or at least to both of them. Examples of node communication could be: the nodes attract each other, or they repel each other, they could change color, or size. They could literally pass information – each node could have some property such as “power”, and when they communicated, the one with more power could give some of its power to the other one. This would probably be best represented by some visual clue as well, such as scaling the nodes according to how much power they have.

Again, we’ll go with a very simple communication theme. Communication between two nodes will be indicated by drawing a line between the two. Easy enough! Here is the whole file with our first node garden:

numNodes = 50;

nodez = [];

for(i=0;i<numNodes;i++){


node = attachMovie(“node”, “n”+i, i);


nodez[i] = node;


node._x = Math.random()*550;


node._y = Math.random()*400;


node._xscale=node._yscale=Math.random()*100+50;

}

lineStyle(1, 0xffffff, 100);

minDist = 80;

for(i=0;i<nodez.length-1;i++){


nodeA = nodez[i];


for(j=i+1;j<nodez.length;j++){



nodeB = nodez[j];



dx = nodeB._x – nodeA._x;



dy = nodeB._y – nodeA._y;



dist = Math.sqrt(dx*dx+dy*dy);



if(dist<minDist){




moveTo(nodeA._x, nodeA._y);




lineTo(nodeB._x, nodeB._y);



}


}

}

There we have it. You should have a bunch of nodes appear on screen with a bunch of lines connecting them, as long as they are within 80 pixels of each other.

Varying the Communication

We can make it look a little better, making the brightness of the communication line proportional to the distance. We can do that with the alpha parameter of the lineStyle command. The alpha will vary between 100 (at a distance of zero) and 0 (when it reaches minDist). To do that, we get the percentage of dist to minDist, and multiply times 100. Actually, that winds up giving us the exact opposite effect, so we subtract that number from 100 to get the figure to use for alpha:

alpha = 100-dist/minDist*100;

Say dist is 80.

80/80 = 1

1*100 = 100

100-100 = 0

If dist is 40, 

40/80 = .5

.5*100 = 50

100 – 50 = 50

Perfect. Just what we want. Plugging that in to the whole mess, we get:

numNodes = 50;

nodez = [];

for(i=0;i<numNodes;i++){


node = attachMovie(“node”, “n”+i, i);


nodez[i] = node;


node._x = Math.random()*550;


node._y = Math.random()*400;


node._xscale=node._yscale=Math.random()*100+50;

}

minDist = 80;

for(i=0;i<nodez.length-1;i++){


nodeA = nodez[i];


for(j=i+1;j<nodez.length;j++){



nodeB = nodez[j];



dx = nodeB._x – nodeA._x;



dy = nodeB._y – nodeA._y;



dist = Math.sqrt(dx*dx+dy*dy);



if(dist<minDist){




alpha = 100-dist/minDist*100;




lineStyle(1, 0xffffff, alpha);




moveTo(nodeA._x, nodeA._y);




lineTo(nodeB._x, nodeB._y);



}


}

}

This should give you a much nicer display, such as this:

Animated Nodes

OK, so far we have a pretty static picture. Let’s animate it a bit. We do that by giving each node a function to run on onEnterFrame. We’ll call that function “nodeMove”. It will be responsible for moving the node around the screen. We’ll also assign each node a random vx and vy property. These give it a velocity on each axis. Finally, we will put the communication nested loop code in a function assigned to _root.onEnterFrame. This will cause the nodes to constantly update their relationships and relative communication. Here’s what all that gives us:

numNodes = 25;

nodez = [];

for(i=0;i<numNodes;i++){


node = attachMovie(“node”, “n”+i, i);


nodez[i] = node;


node._x = Math.random()*550;


node._y = Math.random()*400;


node._xscale=node._yscale=Math.random()*100+50;


node.vx = Math.random()*10-5;


node.vy = Math.random()*10-5;


node.onEnterFrame = nodeMove;

}

minDist = 120;

_root.onEnterFrame = function(){


var i, j, nodeA, nodeB, dx, dy, dist, alpha;


clear();


for(i=0;i<nodez.length-1;i++){



nodeA = nodez[i];



for(j=i+1;j<nodez.length;j++){




nodeB = nodez[j];




dx = nodeB._x – nodeA._x;




dy = nodeB._y – nodeA._y;




dist = Math.sqrt(dx*dx+dy*dy);




if(dist<minDist){





alpha = 100-dist/minDist*100;





lineStyle(1, 0xffffff, alpha);





moveTo(nodeA._x, nodeA._y);





lineTo(nodeB._x, nodeB._y);




}



}


}

}

function nodeMove(){


this._x += this.vx;


this._y += this.vy;


if(this._x>550){



this._x = 0;


} else if(this._x<0){



this._x = 540;


}


if(this._y>400){



this._y = 0;


} else if(this._y<0){



this._y = 400;


}

}

A couple of things to note here. First, notice that I declared all the variables for the _root.onEnterFrame function right at the top. Having local variables in a function keeps them off the main time line and tends to speed things up a bit as they are right at the end of the scope chain – the first place that Flash looks for a variable in a function is the local variables for that function. That’s where they are, so it doesn’t need to look any further.

Next, we clear all previous communication lines by using the clear() command of the drawing API.

The nodeMove function just adds velocity to the position of each node. If a node moves off the screen, it wraps it around to the other side. I hard coded in the values for a 540x400 stage size for simplicity. Better coding practice would be to use variables whose values are obtained by Stage.getWidth() and Stage.getHeight(), which would adjust themselves no matter what size stage you have.

I also reduced the number of nodes to 25, to make things run a little smoother, and upped the minimum distance for communication to 120 to make up for the scarcity.

Already, we are getting into some very cool stuff here, in my own opinion. But this is only the very basic beginning. We can add more and different features to just about any aspect of this garden.

Node Attraction

Earlier I mentioned one idea for node communication was to have them attract to each other. Here is a simple example of this using elasticity, as described in one of my other tutorials.

We’ll start with the last file we just made as a base. If the nodes decide to communicate, they will draw a line. In addition, they will spring towards each other.

To form the spring, we will determine the force between the two nodes. Half of that force will be added to one node’s velocity, the other half to the other one. Remember also, that springs need a spring factor, k, and a damping factor as well. We’ll put that right up top. Due to the amount of nodes attracting each other, and the fact that they want to snap right together, I had to use a very low k, and barely any damping at all. The numbers I give here produce an interesting result. Play around with them and see what you can come up with. Too far one way and they collapse in on each other. Too far the other way and the whole system just goes haywire.

damp = .999;

k = .001;

numNodes = 25;

nodez = [];

for(i=0;i<numNodes;i++){


node = attachMovie(“node”, “n”+i, i);


nodez[i] = node;


node._x = Math.random()*550;


node._y = Math.random()*400;


node._xscale=node._yscale=Math.random()*100+50;


node.vx = Math.random()*10-5;


node.vy = Math.random()*10-5;


node.onEnterFrame = nodeMove;

}

minDist = 120;

_root.onEnterFrame = function(){


var i, j, nodeA, nodeB, dx, dy, dist, alpha, ax, ay;


clear();


for(i=0;i<nodez.length-1;i++){



nodeA = nodez[i];



for(j=i+1;j<nodez.length;j++){




nodeB = nodez[j];




dx = nodeB._x – nodeA._x;




dy = nodeB._y – nodeA._y;




dist = Math.sqrt(dx*dx+dy*dy);




if(dist<minDist){





alpha = 100-dist/minDist*100;





lineStyle(1, 0xffffff, alpha);





moveTo(nodeA._x, nodeA._y);





lineTo(nodeB._x, nodeB._y);





ax = dx*k*.5;





ay = dy*k*.5;





nodeA.vx += ax;





nodeA.vy += ay;





nodeB.vx -= ax;





nodeB.vy -= ay;




}



}


}

}

function nodeMove(){


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;


if(this._x>550){



this._x = 0;


} else if(this._x<0){



this._x = 540;


}


if(this._y>400){



this._y = 0;


} else if(this._y<0){



this._y = 400;


}

}
The rest of the major changes occur when the two nodes decide to communicate. This probably needs some explanation. The code is here:

ax = dx*k*.5;

ay = dy*k*.5;

nodeA.vx += ax;

nodeA.vy += ay;

nodeB.vx -= ax;

nodeB.vy -= ay;

First we calculate an acceleration factor on each axis, ax and ay. This is done, per the elasticity tutorial, by taking a portion of the distance to the target, or multiplying the distance by k. We then multiply this by .5, since only half this acceleration will be applied to each node.

Now comes the tricky part – distributing this force. Basically, you need to add the acceleration to one node and subtract it from the other. Which is which depends on which node position you subtracted from which in order to get the distance in the lines:

dx = nodeB._x – nodeA._x;

dy = nodeB._y – nodeA._y;

Here, we subtracted A from B, so we will need to add the resultant acceleration to A in order to push it towards B. Likewise you will need to subtract it from B to send it in A’s direction. If that’s not clear, just try it either way. If they bounce apart, you got it wrong! After a while, you’ll get a feel for it. I’m up to getting it right about 60% of the time on the first try!

On the other hand, you could leave them reversed and have them repel each other.

Don’t forget to damp the acceleration in the nodeMove function too.

This is essentially the first node garden that I made personally. I feel that this in itself is one of the coolest things I’ve made in Flash, and yet it is still only the beginning!

Rather than go through the specifics of additional files, I’d rather just give you some ideas on other stuff I did.

One was to have two types of nodes. For this, I gave the node movie clip two frames. In one I had a red colored node. In the second frame a blue one. While creating the nodes, I randomly chose a number 1 or 2. If 1, I would tell the node node.gotoAndStop(1) and assign a property node.color = “red”. If 2, I would send it to frame 2 and set the color property to “blue”. Then, in the decision process, I compared colors, something like:

if(nodeA.color == nodeB.color){

If they were the same, they would attract. If different, they would repel. Or vice versa. This would probably be combined with a distance check as well.

Another version I made set a “size” property for every node, at something like 50. Then, every time two nodes communicated, their sizes would be increased by say 5. Then, in the nodeMove function, the _xscale and _yscale of the node would be set to the size factor. Then size would be set back to 50. The result was that the more connections a node had, the bigger it would get. Nodes with no connections would be small. The same thing could be done using _alpha instead.

You can also play around with node creation and destruction. The code I gave you allows for this without a hitch, as long as all nodes are added to the nodez array. You can add nodes by using nodez.push(newNode),  where newNode contains a reference to the newly created node. To remove a node, you need to first remove it from the array, then remove the movie clip. To remove it from the array, you would probably use the splice command. But you would need to know the index of the element where the node is stored. If you know that, you can say simply:

deadNode = nodez[index];

nodez.splice(index, 1);

deadNode.removeMovieClip();

The second line says to start at index, and remove 1 element from the array. 

If you don’t know the index, but you have a reference to the clip, you can loop through the array, testing to see if the current element is the node you are looking for. For example, say you wanted to do this from within the nodeMove function. The key word, this, refers to the current node, but you don’t know where it is in the array. You could do this:

// this is inside the nodeMove function:

if(someCondition){


for(i=0;i<nodez.length;i++){



if(nodez[i] == this){




nodez.splice(i, 1);



}


}


this.removeMovieClip();

}

Some other ideas: You can have the nodes interact not only with other nodes, but with other objects on the stage, or with the mouse. I could go on and on. If you look through my site, www.bit-101.com, you will see several examples of these node ideas, and I myself have only scratched the surface of the iceberg.

Again, thanks to Jared Tarbell for inspiring me in the first place. Let us both know if you come up with anything particularly interesting!

Keith Peters

